The background is a dark blue image featuring various business and data-related icons. These include a dollar sign, a refresh symbol, a hand pointing at a screen, a pie chart with a 12% label, a bar chart, a line graph, a key, a scale of justice, and a network of hexagons. The text is overlaid on this background.

MICROSOFT EXCEL

Продвинутый пользователь Excel

Синтаксис формул

Все формулы начинаются со знака равенства

Аргументы формулы находятся внутри круглых скобок

В русской версии Excel аргументы разделяются точкой запятой (;), но другие регионы могут использовать иные разделители

`=СУММЕСЛИ(`
`СУММЕСЛИ(диапазон; критерий; [диапазон_суммирования])`

Название функции сообщает Excel, какой тип операции вы собираетесь выполнить (Excel предлагает ~ 500 функций)

Примечание: Имена функций не чувствительны к регистру и не всегда обязательны; для базовых арифметических и логических операции часто не нужны:

`= A1 + B1`
`= A1 / B1`
`= A1 > B1`
`= A1 = B1`


Аргументы, зависят от функции и предоставляют Excel информацию, необходимую для расчетов

Примечание: Не все аргументы обязательны. Необязательные аргументы помещены в квадратные скобки [].

У большинства функций есть хотя бы один обязательный аргумент, но некоторым он не нужен, например `СТРОКА()`, `СТОЛБЕЦ()`, `СЕГОДНЯ()`

Абсолютные и относительные ссылки

Существует два **типа ссылок** на ячейки: **относительные** и **абсолютные**. Относительные и абсолютные ссылки ведут себя по-разному при копировании и заполнении в другие ячейки. Относительные ссылки меняются при копировании формулы в другую ячейку. Тогда как, абсолютные ссылки остаются неизменными независимо от того, куда они копируются.

Для копируемой формулы:	Если ссылка:	То меняется на:
	\$A\$1 (абсолютная колонка и строка)	\$A\$1 (ссылка абсолютная)
	A\$1 (относительная колонка и абсолютная строка)	C\$1 (ссылка смешанная)
	\$A1 (абсолютная колонка и относительная строка)	\$A3 (ссылка смешанная)
	A1 (относительная колонка и строка)	C3 (ссылка относительная)

Символ **\$** используется для создания **абсолютных** ссылок. Вы можете зафиксировать всю ячейку (\$A\$1) или только столбец (\$A1) или строку (A\$1), что **предотвращает изменение ссылок** при копировании формулы в новые ячейки.

Для быстрого переключения между типами ссылок используйте клавишу F4

Функции СЧЁТЕСЛИ и СЧЁТЕСЛИМН

Функция **СЧЁТЕСЛИ** позволят посчитать количество значений в диапазоне на основании **ОДНОГО** критерия, **СЧЁТЕСЛИМН** позволят посчитать количество значений в диапазоне на основании **НЕСКОЛЬКИХ** критериев.

=СЧЁТЕСЛИ(диапазон; критерий)

Какие ячейки должны соответствовать вашим критериям?

При каких условиях я хочу подсчитывать?

=СЧЁТЕСЛИМН(диапазон1; критерий1; [диапазон2; критерий2]...)

	A	B	C	D	E	F	G
1	Месяц	Работник	Значение		Критерий	Формула	Результат
2	Январь	Петров	12		Иванов	=СЧЁТЕСЛИ(B2:B11;"Иванов")	3
3	Январь	Сидоров	52		>20	=СЧЁТЕСЛИ(C2:C11;">20")	7
4	Январь	Иванов	54		>20, <50	=СЧЁТЕСЛИМН(C2:C11;">20";C2:C11;"<50")	2
5	Февраль	Иванов	12		Петров, >20	=СЧЁТЕСЛИМН(B2:B11;"Петров";C2:C11;">20")	1
6	Февраль	Сидоров	95				
7	Февраль	Петров	64				
8	Февраль	Сидоров	47				
9	Март	Иванов	26				
10	Март	Петров	2				
11	Март	Сидоров	98				



Аргумент **критерий** должен быть заключен в кавычки, если указывается явно, внутри формулы

Функции СУММЕСЛИ и СУММЕСЛИМН

Функция **СУММЕСЛИ** позволят суммировать значения в диапазоне на основании **ОДНОГО** критерия, **СУММЕСЛИМН** позволят суммировать значения в диапазоне на основании **НЕСКОЛЬКИХ** критериев.

=СУММЕСЛИ (**диапазон**; **критерий**; [**диапазон_суммирования**])

В каких ячейках искать условия?

Какие ячейки суммировать?

При каких условиях я хочу суммировать?

=СУММЕСЛИМН (**диапазон_суммирования**; **диапазон_условия1**; **условие1**; [**диапазон_условия2**; **условие2**]...)

Для указания дат в качестве критерия суммирования, воспользуйтесь функцией ДАТА()



Аргументы функции		
СУММЕСЛИ		
Диапазон	D2:D14	= {42577;42597;42580;42626;42625;4...}
Критерий	">=&ДАТА(2016;9;26)"	= ">=42639"
Диапазон_суммирования	C2:C14	= {1482;1527;4621;2366;2434;3297;27...}

☞ Для функции **СУММЕСЛИ**: если аргумент **диапазон_суммирования** опущен, то аргумент **диапазон** выступает в качестве диапазона условия и диапазона суммирования.

☞ Аналогично функциям подсчета значений и суммирования, существуют функции подсчета среднего значения **СРЗНАЧЕСЛИ()** и **СРЗНАЧЕСЛИМН()**. Синтаксис аналогичен функциям суммирования.

Функция ЕСЛИ

=ЕСЛИ (лог_выражение; [значение_если_истина]; [значение_если_ложь])

Проверяет выражение,
результатом которого может
быть ИСТИНА или ЛОЖЬ
(напр., A1="Иван", A1<B2)

Значение, которое вернет
формула, если логическое
выражение будет ИСТИНА

Значение, которое вернет
формула, если логическое
выражение будет ЛОЖЬ

Год	А	В	С
	Дата	Золото, руб/гр	Подорожало?
1	06.04.2021	4 251.11	
2	03.04.2021	4 221.60	=ЕСЛИ(B3>B2;"Да";"Нет")
3	02.04.2021	4 181.97	Нет
4	01.04.2021	4 098.42	Нет
5	31.03.2021	4 129.57	Да
6	30.03.2021	4 202.90	Да
7	27.03.2021	4 208.46	Да

=ЕСЛИ(B3>B2;"Да";"Нет")

В этом примере, функция ЕСЛИ возвращает ДА
(Подорожало), если значение текущего дня больше
предыдущего, в противном случае НЕТ

Вложение функции ЕСЛИ

Используя **вложенные операторы ЕСЛИ**, вы можете включать несколько логических проверок условий

D3					=ЕСЛИ(В3>В2;ЕСЛИ((В3-В2)>50;"Внимание";"Да");"Нет")
	A	B	C	D	E
1	Дата	Золото, руб/гр	Подорожало?	С вложением	
2	02.03.2021	4 058.78			
3	03.03.2021	4 144.35	Да	Внимание	
4	04.03.2021	4 082.19	Нет	Нет	
5	05.03.2021	4 056.73	Нет	Нет	
6	06.03.2021	4 058.47	Да	Да	
7	10.03.2021	4 065.79	Да	Да	

=ЕСЛИ(В3>В2;ЕСЛИ((В3-В2)>50;"Внимание";"Да");"Нет")

- Если цена подорожала более, чем на 50 руб., то «Внимание»
- Если цена подорожала менее, чем на 50 руб., то «Да»
- Если цена подешевела, то «Нет»

Функция ВПР

Ищет **значение** в крайнем левом столбце **таблицы** и возвращает значение ячейки, находящейся в указанном **столбце** той же строки. ;

= ВПР (искомое_значение; таблица; номер_столбца; [интервальный_просмотр])

Это значение, которое вы пытаетесь найти в таблице

Это там ГДЕ вы пытаетесь
найти искомое значение

В каком столбце содержатся данные, которые вы ищете?

Вы пытаетесь найти точное совпадение (0) или что-то подобное (1)?

B10				=ВПР(A10;A2:C6;3;ЛОЖЬ)
	A	B	C	
1	Табельный №	Имя работника	Отдел	
2	1059236	Кевин Костнер	Бухгалтерия	
3	1093091	Брюс Уиллис	Отдел кадров	
4	1099845	Памелла Андерсон	Производство	
5	1037734	Сальма Хаек	Логистика	
6	1084228	Колин Фарел	Метрология	
7				
8				
9	Табельный №	Отдел		
10	1037734	Логистика		

= ВПР (A10; A2:C6; 3; ЛОЖЬ)

Функция ВПР ищет значение **1037734** в первом столбце (столбце А) таблицы в диапазоне **A2:C6** и возвращает значение **Логистика**, найденное в третьем столбце (столбце С) этой таблицы.

При значении ЛОЖЬ функция возвращает точное совпадение.

Функция ГПР

Ищет **значение** в крайней верхней строке **таблицы** и возвращает значение ячейки, находящейся в указанной **строке** того же столбца.

= ГПР (искомое_значение; таблица; номер_строки; [интервальный_просмотр])

Это значение, которое вы пытаетесь найти в таблице

Это там ГДЕ вы пытаетесь найти искомое значение

В какой строке содержатся данные, которые вы ищете?

Вы пытаетесь найти точное совпадение (0) или что-то подобное (1)?

B7	=HLOOKUP(A7;\$B\$1:\$F\$3;3;0)					
	A	B	C	D	E	F
1	Табельный №	1059236	1093091	1099845	1037734	1084228
2	Имя работника	Кевин Костнер	Брюс Уиллис	Амелла Андерсс	Сальма Хаек	Колин Фарел
3	Отдел	Бухгалтерия	Отдел кадров	Производство	Логистика	Метрология
4						
5						
6	Табельный №	Отдел				
7	1037734	Логистика				

= ГПР (A7; B1:F3; 3; ЛОЖЬ)

Функция ГПР ищет значение **1037734** в первой строке (строке 1) таблицы в диапазоне **B1:F3** и возвращает значение **Логистика**, найденное в третьем строке (строка 3) этой таблицы.

При значении **ЛОЖЬ** функция возвращает точное совпадение.

Функция ИНДЕКС

Функция **ИНДЕКС** возвращает конкретное значение ячейки в массиве на основе заданной строки и столбца

= ИНДЕКС (массив; номер_строки; [номер_столбца])

В каком диапазоне
ячеек вы ищете?

На сколько строк вниз находится
ваше искомое значение?

На сколько столбцов вправо
находится ваше искомое значение?

	A	B	C	D
1	Линия	Наименование линии	Эффект-ть, %	Загружен- ность, %
2	1	Производство какао-порошка	75%	56%
3	2	Производство шоколадной массы	91%	72%
4	3	Производство начинки	95%	34%
5	4	Производство шоколадных плиток	78%	89%
6	5	Упаковочная линия 1	82%	95%
7	6	Упаковочная линия 2	74%	92%
8	7	Палетайзер	92%	32%

=ИНДЕКС(\$A\$1:\$D\$8; 5; 4) = **89%**

Функция ИНДЕКС ищет в диапазоне **\$A\$1:\$D\$8** значение, находящееся на **5 строк** ниже и **4 строки** правее, и возвращает значение **89%**

Функция ПОИСКПОЗ

Функция **ПОИСКПОЗ** возвращает позицию конкретного значения в одномерном вертикальном или горизонтальном массиве.

= ПОИСКПОЗ (искомое_значение; просматриваемый_массив; [тип_сопоставления])

Это значение, которое вы пытаетесь найти в массиве

Диапазон ячеек ГДЕ мы ищем
необходимое нам значение (это должен
быть одномерный массив)

Вы пытаетесь найти точное совпадение (0) или что-то подобное?

1: Ищет наибольшее значение $<$ или $=$ искомому значению

0: Ищет точное совпадение искомому значению

-1: Ищет наименьшее значение \geq или $=$ искомому значению

год	:	✕	✓	f_x	=ПОИСКПОЗ(G2;B1:B8;0)			
	A	B	C	D	E	F	G	H
	Линия	Наименование линии	Эффек- ть, %	Загружен- ность, %				
1	1	Производство какао-порошка	75%	56%				
2	2	Производство шоколадной массы	91%	72%				
3	3	Производство начинки	95%	34%				
4	4	Производство шоколадных плиток	78%	89%				
5	5	Упаковочная линия 1	82%	95%				
6	6	Упаковочная линия 2	74%	92%				
7	7	Палетайзер	92%	32%				
8								
9								

Функция ПОИСКПОЗ ищет в массиве **B1:B8** **Точное совпадение** значения **Упаковочная линия 1**, и возвращает значение **6**

Аналогично работает и с горизонтальными массивами

Комбинирование функций ИНДЕКС и ПОИСКПОЗ

Связка функции **ИНДЕКС** и **ПОИСКПОЗ** часто используется для создания динамической формулы подстановки

G4		=INDEX(B2:D6;MATCH(G2;A2:A6;0);MATCH(G3;B1:D1;0))					
	A	B	C	D	E	F	G
1	Страна	Цена	Курс	ВВП			
2	Бразилия	22	5,5	8751		Страна	Китай
3	Китай	22	6,5	10287		Показатель	ВВП
4	Норвегия	52	8,5	75294		Значение	10287
5	Россия	135	74,6	11601			
6	США	6	1,0	65254			
7							

Данная функция возвращает значение из массива **B2:D6**, заданное определенной строкой и столбцом.

Значение строки и столбца помогают найти функции ПОИСКПОЗ

=ИНДЕКС(B2:D6; ПОИСКПОЗ(G2;A2:A6;0); ПОИСКПОЗ(G3;B1:D1;0))

Массив, из которого формула возвращает значение искомой ячейки

Возвращает позицию найденной страны (Китай) в диапазоне A2:A6 **=2**

Возвращает позицию найденного показателя (ВВП) в диапазоне B1:D1 **=3**

Функция СМЕЩ

Функция **СМЕЩ** возвращает ссылку на диапазон с определенной отправной точкой с заданной высотой и шириной.

=СМЕЩ(ссылка; смещение по строкам; смещение по столбцам; [высота]; [ширина])

Начальная
точка

На сколько строк
вверх или вниз вы
должны сместиться
относительно
начальной точки

На сколько столбцов
вправо или влево вы
должны сместиться
относительно начальной
точки

Если требуется вернуть
диапазон, указываем его
высоту и ширину



Часто применяется в связке с функцией **СЧЁТЗ** для создания динамических диапазонов.

Пример формулы расчета среднего значения последних семи элементов в колонке A:

=СРЗНАЧ(СМЕЩ(A1;СЧЁТЗ(A:A)-7;0;7;1))

Примеры формул, результат и описание

	A	B	C	D	E	F	G	H	I	J	K	L	M	N
1												Формула	Ссылка	Описание
2												=СМЕЩ(A1;0;0;4;3)	A1:C4	Возвращает диапазон состоящий из 4-х строк и 3-х столбцов, начало в A1
3												=СМЕЩ(A1;0;0;3;4)	A1:D3	Диапазон состоящий из 3-х строк и 4-х столбцов
4												=СМЕЩ(A1;1;1;4;3)	B2:D5	Смещается на 1 строку вниз и 1 колоку вправо, диапазон 4x3
5												=СМЕЩ(A1;1;2;5;6)	C2:H6	Смещается на 1 строку вниз и 2 колоки вправо, диапазон 5x6
6														
7														
8														

Функция СУММПРОИЗВ

Функция **СУММПРОИЗВ** возвращает сумму произведений из соответствующих диапазонов или массивов (Все массивы должны быть одинаковых размеров)

=СУММПРОИЗВ (массив1; массив2 ... массив_N)

Без использования СУММПРОИЗВ

D7				
	A	B	C	D
1	Товар	Цена за ед	Количество	Итого
2	Зеленый чай	58 Р	9	522
3	Черный чай	49 Р	7	343
4	Мате	124 Р	3	372
5	Фиточай	135 Р	5	675
6				
7		Общие продажи		1912

Чтобы посчитать общую сумму произведений, необходимо 1) перемножить каждую строчку; 2) суммировать полученные произведения

С использованием СУММПРОИЗВ

C7				
	A	B	C	D
1	Товар	Цена за ед	Количество	
2	Зеленый чай	58 Р	9	
3	Черный чай	49 Р	7	
4	Мате	124 Р	3	
5	Фиточай	135 Р	5	
6				
7		Общие продажи		1912

=СУММПРОИЗВ (B2:B5;C2:C5) =1912

Делает те же операции за один шаг

Функции НАИБОЛЬШИЙ/НАИМЕНЬШИЙ, РАНГ

	A
1	Значение
2	39
3	78
4	88
5	72
6	38
7	56
8	49
9	50

Функции **НАИБОЛЬШИЙ/НАИМЕНЬШИЙ** возвращают n-ое наибольшее или наименьшее значение массива

=НАИБОЛЬШИЙ (A2:A9; 2) = 78

Возвращает 2-е наибольшее значение из массива A2:A9

=НАИМЕНЬШИЙ (A2:A9; 3) = 49

Возвращает 3-е наименьшее значение из массива A2:A9

	A
1	Значение
2	39
3	78
4	88
5	72
6	38
7	56
8	49
9	50

- =РАНГ (A2; A2:A9) = 7
- =РАНГ (A3; A2:A9) = 2
- =РАНГ (A4; A2:A9) = 1 (наибольший)
- =РАНГ (A5; A2:A9) = 3
- =РАНГ (A6; A2:A9) = 8 (наименьший)
- =РАНГ (A7; A2:A9) = 4
- =РАНГ (A8; A2:A9) = 6
- =РАНГ (A9; A2:A9) = 5

Функция **РАНГ** возвращает ранг (величину относительно других значений в списке) числа в списке чисел

Формулы массивов - Введение

Формула массива — это формула, которая может выполнять несколько вычислений с одним или несколькими элементами массива. Формулы массива могут возвращать как несколько результатов (Формула массива с несколькими ячейками), так и один (Формула массива с одной ячейкой).

Чтобы ввести, редактировать или удалить формулу массива, необходимо нажать **CTRL-SHIFT-ENTER**; при этом автоматически добавляются фигурные скобки «{}», которые указывает, что данная формула является формулой массива.

	A	B	C	D
1	Месяц	Цена	Количество	
2	Январь	97,0 ₽	5	485,0 ₽
3	Февраль	69,0 ₽	7	
4	Март	64,0 ₽	1	
5	Апрель	51,0 ₽	5	

Если выбрать диапазон **D2:D5**, ввести **=B2:B5*C2:C5** и нажать **ENTER**, формула будет применена только к ячейке **D2**

	A	B	C	D
1	Месяц	Цена	Количество	
2	Январь	97,0 ₽	5	485,0 ₽
3	Февраль	69,0 ₽	7	483,0 ₽
4	Март	64,0 ₽	1	64,0 ₽
5	Апрель	51,0 ₽	5	255,0 ₽

Если выбрать диапазон **D2:D5**, ввести **=B2:B5*C2:C5** и нажать **CTRL-SHIFT-ENTER**, будет создана формула массива, которая будет применена для всего диапазона **D2:D5**

Формулы массивов - Правила

При работе с **формулами массива** необходимо соблюдать следующие правила:



1. Чтобы ввести или отредактировать формулу массива, необходимо ввести сочетание клавиш **CTRL-SHIFT-ENTER**
2. Для формулы массива с несколькими ячейками необходимо выбрать диапазон ячеек *перед* вводом формулы.
3. Вы не можете изменить содержимое отдельной ячейки, которая является частью формулы массива.
4. Вы можете переместить или удалить всю формулу массива, но не ее часть (поэтому вам часто придется удалять и переделывать)
5. Вы не можете вставлять пустые ячейки или удалять ячейки из формулы массива с несколькими ячейками.

Формулы массивов – ЗА и ПРОТИВ

Формулы массивов невероятно мощный инструмент, однако с ними бывает очень сложно работать. Вот несколько ЗА и ПРОТИВ использования формул массивов

ЗА


- Позволяет объединить несколько вычислений в одной формуле, часто сокращая размер файла
- Позволяет выполнять сложные вычисления, которые обычные функции сделать не могут
- Сокращает риск человеческой ошибки таких как: случайное удаление части массива или неправильный ввод формулы

ПРОТИВ

- Сложно изменять или удалять существующие формулы массивов
- Ограниченная видимость работы функции, особенно для пользователей, которые не знакомы с массивами
- Отсутствует возможность изменения ячеек внутри массивов
- Может снизить скорость работы файла, если используется несколько формул массивов


Размерность массивов

Константы массивов создаются путём ручного ввода элементов массива напрямую в панель формул и ручного ввода фигурных скобок {}.




	A	B	C	D
1	1	2	3	4

Константа горизонтального массива создает массив, состоящий из одной строки и разделенный точкой с запятой* (напр, выделить диапазон A1:D1, ввести “={1;2;3;4}” и нажать C-S-E)



	A
1	1
2	2
3	3
4	4

Константа вертикального массива создает массив, состоящий из одного столбца и разделенный двоеточием* (напр, выделить диапазон A1:A4, ввести “={1:2:3:4}” и нажать C-S-E)



	A	B	C	D
1	1	2	3	4
2	5	6	7	8

Константа двухмерного массива создает массив, состоящий как из столбцов, так и строк (напр, выделить диапазон A1:D2, ввести “={1;2;3;4;5;6;7;8}” и нажать C-S-E)



* ВНИМАНИЕ!!! Разделители могут отличаться в зависимости от настроек вашей системы

Константы массивов в формулах

Константы массивов могут содержать числовые значения, текст (окруженный кавычками “”), логические значения (ИСТИНА, ЛОЖЬ), ошибки (#Н/Д) и могут быть использованы как часть формул массивов.

$$\{=СУММ (A1:A5 * \{1:2:3:4:5\})\} = 82$$

- 1 В памяти компьютера создается вертикальный массив

1
2
3
4
5

- 2 Значения диапазона A1:A5 перемножаются с вертикальным массивом

	A
1	4
2	8
3	7
4	9
5	1

x

1
2
3
4
5

- 3 В результате умножения возникает массив с произведениями

4
16
21
36
5

- 4 Элементы данного массива суммируются

4
16
21
36
5


82




При вводе константы массива фигурные скобки проставляются вручную, однако еще одни фигурные скобки будут добавлены автоматически, как только на нажмете **CTRL-SHIFT-ENTER**

Именованные константы массивов


Именованные константы массивов задаются точно также, как обычные именованные диапазоны - с помощью Диспетчера имен, что упрощает их использование



Во вкладке **Формулы** выбираем **Диспетчер Имен** -> **Новый**



Чтобы константа массива появилась на рабочем листе, выделите диапазон **A1:C1**, введите «**=Квартал1**» и нажмите сочетание клавиш **CTRL-SHIFT-ENTER**



В диалоговом окне **Создание имени** прописываем константу массива (фигурные скобки проставляются вручную), задаем Имя и жмем **OK**

Функция ТРАНСП

Функция ТРАНСП позволяет изменять ориентацию таблицы так, чтобы строки стали столбцами, а столбцы строками. Функция преобразует вертикальный диапазон ячеек в горизонтальный и наоборот.

	A	B	C	D	E
1	1	2			
2	3	4			
3	5	6			
4	7	8			
5	9	10			
6					
7	1	3	5	7	9
8	2	4	6	8	10
9					

{=ТРАНСП (A1:B5)}

Формула транспонирования в качестве аргумента использует диапазон **A1:B5** и преобразует строки в столбцы, а столбцы в строки.

Выделите диапазон **A7:E8**, введите формулу «**=ТРАНСП(A1:B5)**» и нажмите сочетание клавиш **CTRL-SHIFT-ENTER**

СОВЕТ:

Чтобы скопировать и вставить диапазон в транспонированном виде,

- 1) Скопируйте исходный диапазон в буфер обмена
- 2) Выберите Специальная вставка -> Транспонировать



Ваш первый макрос

Знакомство с макросами и редактором Visual Basic

- 1 Как записать макрос
- 2 Знакомство с редактором Visual Basic
- 3 Объектно-ориентированная модель кода VBA

Объектно-ориентированная модель VBA



Application (Приложение)



Workbooks (Рабочие книги)



Workbook (Рабочая книга)



Worksheets (Рабочие листы)



Worksheet (Рабочий лист)



Range (Диапазон)

Примеры:

Application.Workbooks("КНИГА1.XLSX")

Application.Workbooks("КНИГА1.XLSX").Worksheets("Лист1")

Application.Workbooks("КНИГА1.XLSX").Worksheets("Лист1").Range("A1")

Worksheets("Лист1").Range("A1")

Range("A1")



Свойства объектов

Свойство — это атрибут объекта, которой определяется одно из характеристик объекта, такая как размер, цвет, местоположение экрана или характер поведения объекта, например включен ли он и является ли видимым. Чтобы изменить характеристики объекта, изменяют значения его свойств.



Марка

Модель

Поколение

Цена
Цена от до, руб.

Тип автомобиля
☒ Все ☐ С пробегом ☐ Новые

Год выпуска
от 1960 до 2021

Коробка передач
☐ автомат ☐ вариатор

Свойства Range

- AddIndent
- Address
- AddressLocal
- AllowEdit
- Application
- Areas
- Borders
- Cells
- Characters
- Column
- Columns
- ColumnWidth
- Comment
- CommentThreaded
- Count
- CountLarge
- Creator
- CurrentArray
- CurrentRegion
- Dependents
- DirectDependents
- DirectPrecedents
- DisplayFormat
- End
- EntireColumn
- EntireRow
- Errors
- Font

```
Worksheets ("Лист 1").Range ("A1").Value
```

Чтобы обратиться к свойству, нужно сначала **обратиться к объекту**, а затем к **свойству** этого объекта, используя **точку** в качестве разделителя.

Методы объектов

Метод — это действие, которое может выполняться объектом. Например, Add — это метод объекта ComboBox, так как им добавляется новая запись в поле со списком.



- Двигаться
- Светить фарами
- Сигналить
- Стоять в пробке
- ...

Методы Range

- Activate
- AddComment
- AddCommentThreaded
- AdvancedFilter
- AllocateChanges
- ApplyNames
- ApplyOutlineStyles
- AutoComplete
- AutoFill
- AutoFilter
- AutoFit
- AutoOutline
- BorderAround
- Calculate
- CalculateRowMajorOrder

```
Range("A1").Clear
```

```
Range("A1").Copy Range("A2")
```

Чтобы обратиться к методу, нужно сначала **обратиться к объекту**, а затем к **методу** этого объекта, используя **точку** в качестве разделителя.

